

MetaPost: A Reference Manual

Peter Grogono

Wednesday 9th June, 2010

Contents

1	Introduction	1
2	Basic Concepts	2
2.1	Version	2
2.2	Input format	2
2.3	The L ^A T _E X File	2
2.4	Comments	4
2.5	Units	4
2.6	Coordinates	4
2.7	Variables	5
2.8	Internal Variables	5
2.9	Declarations	6
2.10	Scope	6
2.11	Expressions	7
2.12	Assignment	8
2.13	Equations	8
3	Types	9
3.1	Booleans	9
3.2	Strings	10
3.3	Colors	11
3.4	Numerics	12
3.5	Pairs	13
3.6	Pens	14
3.7	Transforms	17
3.8	Pictures	19
3.9	Lists	20
4	Paths	21
4.1	Straight lines	21
4.2	Dots and Dashes	22
4.3	Circles, Disks, and Arcs	22
4.4	Curves	23
4.5	Parametric Paths	24

Contents

4.6	Path Constructors	27
5	Commands	28
5.1	Drawing Commands	28
5.1.1	<code>btex</code> and <code>verbatimtex</code>	28
5.1.2	<code>clip</code>	29
5.1.3	<code>draw</code>	29
5.1.4	<code>drawarrow</code>	30
5.1.5	<code>fill</code>	30
5.1.6	<code>label</code>	31
5.2	Non-drawing Commands	31
5.2.1	<code>drawoptions</code>	31
5.2.2	<code>filenametemplate</code>	32
5.2.3	<code>for</code>	32
5.2.4	<code>if</code>	33
5.2.5	<code>message</code>	34
5.2.6	<code>readfrom</code>	35
5.2.7	<code>save</code>	35
5.2.8	<code>show</code> and friends	35
5.2.9	<code>write</code>	36
6	Macros	36
6.1	<code>def</code> macros	36
6.2	<code>vardef</code> and other macro forms	38
7	Macro Packages	38
7.1	Boxes	39
7.1.1	Creating and Drawing Boxes	39
7.1.2	Positioning Boxes	40
7.1.3	Oval Boxes	41
7.2	Graphs	42
7.3	TEX	43
8	Debugging	44
9	Examples	44
9.1	Euler Integration	45
9.2	The Lorentz Transformation	46
9.3	Dissipation	48
9.4	Desargues' Theorem	49
9.5	Three Dimensions	50
9.6	Reading a File	52
10	Links	53
	Index	54

List of Figures

1	The effect of <code>linecap</code> and <code>linejoin</code>	16
2	Parametric curves	24
3	L ^A T _E X fonts	28
4	Escape sequences for <code>filenamestemplate</code>	32
5	Using <code>for...within</code> to obtain a partial analysis of a picture.	34
6	Box variables	40
7	Oval variables	42
8	Simulated coin tossing	43
9	Euler Integration	45
10	The Lorentz Transformation	47
11	Point numbering used in Figure 10	47
12	Area reduction in a dissipative system	48
13	Desargue's Theorem	50
14	Programming Desargues' Theorem	51
15	Perspective view of a cube	52

1 Introduction

METAPOST is a picture-drawing language that generates diagrams and pictures in embedded postscript. METAPOST, an extension of Donald Knuth's MetaFont, was designed and implemented by John Hobby at AT&T Bell Laboratories.

The official guide for METAPOST, Hobby's *A User's Manual for METAPOST*, referred to in this manual as UMM, is written in a narrative style that is easy to read but hard to learn from. Several other people have written METAPOST manuals in narrative form. This manual is intended both to help beginners and to serve as a reference.

Examples of METAPOST code in this manual are intentionally short. This should make it easier to see the point of the example. With long and complex examples, it is sometimes hard to relate the code to the output. There are some longer examples at the end (§9:44).

This manual is not complete. Important omissions are indicated by a reference to UMM. In particular, UMM includes a complete grammar for METAPOST, whereas this manual relies mostly on examples.


Conventions

- Since there are a large number of cross-references in this manual, we use a concise notation. For example, (§3.7:17) is a reference to Section 3.7 which is to be found on page 17.
- Most lists in METAPOST use comma as a separator. Unless stated otherwise, the term “list of X ” means a list of the form X_1, X_2, X_3, \dots
- In grammar rules, $\langle pred \rangle$ indicates a predicate (boolean-valued expression) and $\langle seq \rangle$ indicates a sequence of commands.
- The command “`show $\langle expr \rangle$` ” displays ‘>>’ followed by the value of the expression (§5.2.8:35). In examples containing `show`, we put the code on the left of the page and the output to the right of it, like this:

<code>show (3,4) + (5,6);</code>	<code>>> (8,10)</code>
----------------------------------	------------------------------

- Similarly, we illustrate drawing techniques by putting code on the left and the picture produced on the right. In these examples, we usually omit the commands `beginfig` and `endfig`, as here:¹

`draw unitsquare rotated 45 scaled 25;`



15

- We use the symbol \triangleq to mean “is defined to be equal to”.

Acknowledgments I am indebted to Brian Shearing for ingenious hints on METAPOST usage — and, indeed, for informing me of its existence in the first place.

¹Numbers in the margin, such as #15 here, are an editing aid and will not appear in the final version of this manual.

2 Basic Concepts

2.1 Version

Features were added to METAPOST as it evolved. This manual describes Version 1.000 but also includes some useful features of later versions. The string constant `mpversion` (§3.2:10) will tell you what version you are using, but only if it is a fairly recent version.

2.2 Input format

METAPOST is a program that reads an input file with a name of the form $\langle name \rangle.mp$ and writes several output files. By default, the names of the output files are $\langle name \rangle.1$, $\langle name \rangle.2$, $\langle name \rangle.3$, etc. For example, if the input file was `fig.mp` then the output files would be `fig.1`, `fig.2`, `fig.3`, and so on. The precise form of the output file name is controlled by the command `filenametemplate` (§5.2.2:32). The output files are read by \TeX , \LaTeX , or any program that accepts embedded postscript files.

In the input file, each figure is introduced by `beginfig(n)`, where n is the number of the figure, and terminated with `endfig`. Commands may occur outside figures as well as inside them. An input file has the general form:

```

 $\langle seq \rangle$ 

beginfig(1);
     $\langle seq \rangle$ 
endfig;

 $\langle seq \rangle$ 

beginfig(2);
     $\langle seq \rangle$ 
endfig;

....

beginfig( $n$ );
     $\langle seq \rangle$ 
endfig;

end

```

2.3 The \LaTeX File

There are several ways of including METAPOST diagrams in a \LaTeX files. Two of them are described below. Assume that the file generated by METAPOST for figure n is called $\langle name \rangle.\langle number \rangle$, where

- $\langle name \rangle.mp$ is the name of the file processed by METAPOST

- $\langle number \rangle$ is the number of the figure

Using the `graphicx` Package Include the `graphicx` package by writing

```
\usepackage{graphicx}
```

in the preamble of your \LaTeX document. To insert METAPost figure n in your document, write

```
\includegraphics{\langle name \rangle.\langle number \rangle}
```

Use optional arguments to `\includegraphics` to scale the diagram if necessary. Use standard \LaTeX commands to position it. One possibility is to use an equation environment to centre the diagram. For example:

```
$$ \includegraphics[width=5in]{mypics.7} $$
```

\LaTeX may not accept a file name with a numerical extension. As mentioned above, however, METAPost can write files with arbitrary names (§5.2.2:32).

Using the `hyperref` Package Include the `hyperref` package by writing

```
\usepackage[...]{hyperref}
```

in the preamble of your \LaTeX document. This package has many useful options, indicated here by ‘...’. To insert METAPost figure n in your document, write

```
\convertMPtoPDF{\langle name \rangle.\langle number \rangle}{\langle Xscale \rangle}{\langle Yscale \rangle}
```

in which:

- $\langle Xscale \rangle$ is the horizontal magnification factor
- $\langle Yscale \rangle$ is the vertical magnification factor

For example, to include Figure 3 generated from `figs.mp` without scaling, write:

```
\convertMPtoPDF{figs.3}{1}{1}
```

The file `mproof.tex` The \TeX file `mproof` creates a `.dvi` file from METAPost output. After the following run, `mproof.dvi` contains the text “figs.65” and the corresponding figure.

```
D:\Pubs>tex mproof figs.65
This is TeX, Version 3.141592 (MiKTeX 2.6)
(figs.65
figs.65: BoundingBox: llx = -1 lly = -3 urx = 61 ury = 72
figs.65: scaled width = 62.23178pt scaled height = 75.28038pt
)
[1]
Output written on mproof.dvi (1 page, 396 bytes).
```

2.4 Comments

METAPOST comments are delimited by the character “%” and the end of the line on which % appears, as in L^AT_EX.

2.5 Units

The default units of METAPOST are *Postscript points*. Postscript points are also called “big points” because they are slightly bigger than printer’s points. The abbreviations for Postscript points and printer’s points are **bp** and **pt**, respectively, as in T_EX:


$$\begin{aligned} 1 \text{ bp} &= 1/72 \text{ inches} \\ 1 \text{ pt} &= 1/72.27 \text{ inches} \end{aligned}$$

METAPOST also understands the other units provided by T_EX: **cm** (centimetres), **mm** (millimetres), **in** (inches), and **pc** (picas: $1\text{pc} = \frac{1}{6}\text{in}$). The units behave like constants defined with the appropriate values. For example, the value of `10*mm` is the length 10 millimetres expressed in Postscript points. The multiplication operator, `*`, can be omitted if the left operand is a number and the right operand is a variable. At the last line, METAPOST does not report an error for `n mm`, but parses it as a compound name (§2.7:5), `n.mm`, with no defined value.

<code>n = 10;</code>	
<code>show 10*mm;</code>	<code>>> 28.3464</code>
<code>show 10mm;</code>	<code>>> 28.3464</code>
<code>show n*mm;</code>	<code>>> 28.3464</code>
<code>show n mm;</code>	<code>>> n.mm</code>

It is sometimes convenient to define a unit and use it to scale every number. Here is an example of picture scaling:

```
u = 1cm;
draw (0u,0u) -- (1.5u,0u) -- (1.5u,1.5u) -- (0u,1.5u) -- cycle;
z1 = (0u,-0.2u);
z2 = (1.5u,-0.2u);
drawdblarrow z1 -- z2;
label.bot("1.5cm", 0.5[z1, z2]);
```




20

2.6 Coordinates

METAPOST uses the conventional coordinate system of mathematics: *X* values increase across the page to the right, and *Y* values increase up the page.

```
drawarrow (0,0) -- (0,30);
drawarrow (0,0) -- (30,0);
dotlabel.llft(btex (0,0) etex, (0,0));
label.rt(btex X etex, (30,0));
label.top(btex Y etex, (0,30));
```



22

Usually, it does not matter where the origin is, because the Postscript file that METAPost generates defines a bounding box for all of the visible components of the picture. The origin might be centered, at the lower left, somewhere else inside the bounding box, or even outside it. However, if the origin is inside the picture, some of the coordinates may be negative. Postscript processors occasionally object to negative coordinates. This problem is rare but, if it occurs, it can be corrected by concluding each figure with the command

```
currentpicture := currentpicture shifted (-llcorner currentpicture);
```

In this command, `currentpicture` (§3.8:19) is an internal variable of type `picture`, `shifted` is a transform (§3.7:17) and `llcorner` gives the coordinates of the lower left corner of a picture.

2.7 Variables

Almost any string to which METAPost does not assign a meaning can be used as a variable name. For example, `@&##$@` is a legal name. In practice, it is usually better to use more conventional names.

In general, a name consists of a *tag* optionally followed by a *suffix*. Any sequence of upper and lower case letters is a valid tag. Tags may also contain other characters, of which the most common is ‘.’. For example, METAPost uses tags such as `b.n` and `b.s` to denote points on boxes.

The suffix may be a simple numeric, as in `p3`, or an expression in brackets, as in `p[i+3]`. Multidimensional indexes must be separated with dots or brackets: `p.3.4` is equivalent to `p[3][4]`.

The names `x`, `y`, and `z` play a special role. First, they are implicitly local (§2.10:6). Second, they are linked by the implicit equations

$$\begin{aligned} z &= (x, y) \\ \text{xpart } z &= x \\ \text{ypart } z &= y \end{aligned}$$

where x , y , and z stand for names that start with `x`, `y`, or `z` and are followed by a suffix. Thus `z.a[3]` denotes the pair $(x.a[3], y.a[3])$.

2.8 Internal Variables

METAPost has a large number of internal variable names. Internal variables are implicitly declared within METAPost but can be given new values by assignment (§2.12:8).

The command `newinternal` introduces new variables that behave like internal variables. This is not very useful in simple diagrams but might be useful in a macro package. To introduce new internal variables `Jack` and `Jill`, write:

```
newinternal Jack, Jill;
```


2.9 Declarations

A type name followed by a list of variable names serves to *declare* the variables named in the list. The type names are: `boolean`, `color`, `numeric`, `pair`, `path`, `pen`, `picture`, `string`, `transform`. Each type is described in detail later (§3:9).

If a variable is used without declaration, it is assumed to be a `numeric`. Occasionally, METAPost is able to determine the type of a variable from its use, and an explicit declaration is not required. In most cases, however, a declaration is required. In particular, a declaration is required when:

- The type of the variable cannot be determined from the context in which it is used.
- A declaration can be used to “undefine” the variable, that is, to change its value to “unknown”.
- An array is being declared. Following the declaration

```
pair p[];
```

the variables `p0`, `p1`, `p2`, ..., `p[i]`, in which `i` is an integer expression, are all available for use.

- Multidimensional arrays can be declared with flexible syntax:

```
path p[][], q[]r[];
```

Corresponding values would include `p[1][2]`, `q[i]r6`, etc.

Declarations with numeric suffixes, such as

```
pair p1, p2;
```

are *not* allowed; the general form `p[]` described above must be used.

2.10 Scope

In general, the scope of a name starts at the point of its declaration or first use and extends to the end of the source file. There are two exceptions to this rule. First, the names `x`, `y`, and `z` play a special role to be described shortly. Second, the constructs `begingroup` and `endgroup` provide an environment in which local names can be declared.

METAPost uses *groups* to provide local variables. The syntax of a group is

```
begingroup <seq> endgroup
```

in which `<seq>` is a sequence of commands. The last item in the sequence may be an expression, in which case the value of this expression becomes the value of the group.

Within a group, the command `save` (§5.2.7:35) followed by a list of variable names makes those names local to the group. The global values of these names, if any, are saved. In the code below:

- `a` is used both globally and locally; its global value, 1, is restored on group exit.
- `b` is defined as a local variable within the group and becomes undefined on group exit.

- `c` is defined within the group but is not declared as a local variable. Consequently, it is treated as a global variable and its value is retained on group exit.

```

a = 1;
begingroup
  save a, b;
  a = 2;
  b = 3;
  c = 4;
endgroup;
show a;
show b;
show c;

```


>> 1
>> b
>> 4

The command `interim` allows you to redefine a internal variable inside a group without changing its global value. For example, the following statements draw an arrow (§5.1.4:30) with an elongated head without changing the global value of `ahlength`:

```

begingroup;
  interim ahlength := 15;
  drawarrow (0,0) -- (40,40);
endgroup;

```



66

The scope of the special names `x`, `y`, and `z` is restricted to a particular figure. That is, the code on the left has the effect of the code on the right (of course, the `beginfig/endfig` environment also creates a figure file, etc.):

<pre> beginfig(n) <seq> endfig </pre>	<pre> begingroup save x, y, z; <seq> endgroup </pre>
---	--

Recommendation: save all of the variables in a figure that are not needed for other figures.

2.11 Expressions

An expression in METAPost is a `<primary>`, `<secondary>`, `<tertiary>`, or `<expression>`. A binary operator with highest precedence is called a `<primary binop>` and its operands must either be `<primary>`s or be enclosed between parentheses. Since unary operators have higher precedence than binary operators, there are four levels of precedence. Binary operators are organized as follows, with the top line having highest priority.

<code><unop></code>	<code>abs angle arclength ...</code>									
<code><primary binop></code>	<code>* / ** and dotprod div mod infont</code>									
<code><secondary binop></code>	<code>+</code>	<code>-</code>	<code>++</code>	<code>+-</code>	<code>or</code>	<code>intersectionpoint</code>	<code>intersectiontimes</code>			
<code><tertiary binop></code>	<code>&</code>	<code><</code>	<code><=</code>	<code><></code>	<code>=</code>	<code>>=</code>	<code>></code>	<code>cutafter</code>	<code>cutbefore</code>	

2 Basic Concepts

However, division (/) has highest precedence if both of its operands are numbers. This means that, for example:

$$\begin{aligned}\text{sqrt } 2/3 &\equiv \sqrt{2/3} \\ \text{sqrt } n/d &\equiv \sqrt{n/d} \\ \text{sqrt}(n/d) &\equiv \sqrt{n/d}\end{aligned}$$

The exponent operator ** has the same precedence as multiply (*). Consequently:

$$3*a**2 \equiv (3a)^2.$$

The meaning of most of the operators are conventional. There are special operators for the hypotenuse and difference of squares:

$$\begin{aligned}a \text{ ++ } b &\equiv \sqrt{a^2 + b^2} \\ a \text{ +-+ } b &\equiv \sqrt{a^2 - b^2}\end{aligned}$$

There are also many operators with names, both unary and binary. The operators that may be used with each type, and their meanings, are discussed with the corresponding types (§3:9).

Some expressions use the keyword `of`. The general format of these expressions is

$$\langle \text{of operator} \rangle \langle \text{expression} \rangle \text{ of } \langle \text{primary} \rangle$$

where $\langle \text{of operator} \rangle$ is one of:

`arctime direction directiontime directionpoint penoffset point`
`postcontrol precontrol subpath substring`

For example (§3.2:10):

$$\text{substring } (2,4) \text{ of "abcde" } \equiv \text{"cd"}.$$

2.12 Assignment

The assignment operator is `:=`, and it has the usual effect of giving the variable on its left side the new value obtained by evaluating its left side. Previous values of the variable are discarded.

Assignment (`:=`) should be distinguished from equality (`=`), which is used in equations, as described in the next section.

2.13 Equations

Variables may receive values either explicitly by assignment, as above, or implicitly by linear equations. Equations use the equality comparison, `=`. Equations may be built up from numerics (§3.4:12), pairs (§3.5:13), or colors (§3.3:11).

MetaPost must be able solve equations before any pictures that use their values are drawn.

3 Types

METAPOST easily solves the equations below, obtaining $a = 3$, $b = 4$, $c = 5$. Note that `2*a` can be abbreviated to `2a`, etc. The effect of the command `showdependencies` is to display the inferences that METAPOST has made from the equations at that point, as shown at the right.

```
2a + 2b + c = 19;
3a-b = 5;
showdependencies;
4b-3c = 1;

>> a=-0.125c+3.625
>> b=-0.375c+5.875
```

A value obtained by solving equations may be changed by assignment. However, the assignment changes *only* the variable assigned.

```
a = 2;           % gives a the value 2
a = b;           % gives b the value 2
a := 3;          % gives a the value 3 but does not change b
```

The symbol `whatever` introduces a new, anonymous variable. It may be used to avoid introducing variables unnecessarily. For example, we could find the intersection `i` of two lines `(p,q)` and `(r,s)` by using the fact that `a[p,q]` is a point on the line from `p` to `q` (§2.11:7).

```
i = a[p,q];
i = b[r,s];
```

Since the names of the numerics `a` and `b` are not needed, we could write instead:

```
i = whatever[p,q] = whatever[r,s];
```

It does not matter that the two instances of `whatever` will have different values.

3 Types

The types provided by METAPOST are: `boolean`, `string`, `color`, `numeric`, `pair`, `path`, `pen`, `picture`, and `transform`. All of them are described in this section except `path`, which has its own section (§4:21).

3.1 Booleans

The values of `boolean` are `true` and `false`. The binary comparison operators `=`, `<>`, `<`, `<=`, `=>`, and `>` return `boolean` values. The unary operator `not` and the binary operators `and` and `or` take `boolean` operands and return the expected `boolean` values. Booleans are also needed for `if` statements (§5.2.4:33).

The equality operators `=` and `<>` work with all types. The comparison operators, `<`, `<=`, `=>`, and `>`, work with most types and have expected values for reasonable types.

3 Types

Type names can be used as predicates:

boolean b	\triangleq	b is a boolean
color c	\triangleq	c is a color
rgbcolor c	\triangleq	c is a rgbcolor
cmykcolor c	\triangleq	c is a cmykcolor
numeric n	\triangleq	n is a numeric
pair p	\triangleq	p is a pair
path p	\triangleq	p is a path
pen p	\triangleq	p is a pen
picture p	\triangleq	p is a picture
string s	\triangleq	s is a string
transform t	\triangleq	t is a transform

Other predicates include:

odd n	\triangleq	the closest integer to n is odd
cycle p	\triangleq	path p is a cycle
known x	\triangleq	x has a value
unknown x	\triangleq	x does not have a value

3.2 Strings

Strings are sequences of characters bounded by double quotes ("). Strings may not contain double quotes or line breaks. The binary comparison operators =, <>, <, <=, =>, and > accept **string** operands and use the underlying character codes for ordering.

Functions for strings include:

string s	\triangleq	true if s is a string
length s	\triangleq	the number of characters in s

Constants and functions yielding strings include:

ditto	\triangleq	the one-character string "
mpversion	\triangleq	a string giving the version of METAPOST
substring (m,n) of s	\triangleq	characters m to n of string s
scantokens (s)	\triangleq	the result of parsing the token sequence s
char n	\triangleq	a string consisting of the character with ASCII code n
decimal n	\triangleq	a string representing the decimal representation of the number n

3 Types

$\text{str } s \triangleq$ the string representation of the suffix s
 $s \& t \triangleq$ the concatenation of strings s and t

The indexing convention for strings is the same as C: the first character has index 0 and the second argument of `substring` indexes the first character following the selection:

`substring (2,4) of "abcde" \equiv "cd".`

The expression

`s infont f`

yields a picture (§3.8:19) consisting of the string s typeset in the font f . There is a `defaultfont` (usually `cmtext10`) with a size `defaultscale` (usually 1).

`draw "MetaPost" infont "cmsl12" scaled 2 rotated 30;`

MetaPost

74

Calling `scantokens` invokes METAPOST's input routine. It can be used as a general conversion function. For example, this call gives `pi` the numerical value 3.14159:

`pi := scantokens("3.14159");`

`scantokens` is one of the few METAPOST operations that creates a list (§3.9:20). See also (§9.6:52) for a more interesting application of `scantokens`.

3.3 Colors

A value of type `color` has three components, corresponding to the red, green, and blue components of a colour. Each component is clamped to the range $[0, 1]$. Colours are written (r, g, b) . METAPOST can solve linear equations involving colours. Two colours may be added or subtracted; a colours may be multiplied and divided by numerics; colours may be used in linear equations.

`show white/4; >> (0.25,0.25,0.25)`
`show 0.2green + 0.6blue; >> (0,0.2,0.6))`

Constants and functions for colours include:

`color c \triangleq` true if c is a RGB colour
`black \triangleq (0,0,0)`
`white \triangleq (1,1,1)`
`red \triangleq (1,0,0)`
`green \triangleq (0,1,0)`
`blue \triangleq (0,0,1)`

3 Types

<code>redpart(<i>r, g, b</i>)</code>	\triangleq	<code>r</code>
<code>greenpart(<i>r, g, b</i>)</code>	\triangleq	<code>g</code>
<code>bluepart(<i>r, g, b</i>)</code>	\triangleq	<code>b</code>
<code>background</code>	\triangleq	the background colour (default = <code>white</code>)

To change the foreground colour (i.e., the drawing colour), use `drawoptions` (§5.2.1:31).

Later versions METAPOST support the CMYK colour model as well as RGB. A CMYK colour is represented as a four-tuple (`c,m,y,k`) with (1,1,1,1) representing black and (0,0,0,0) representing white. The following constants and functions are provided:

<code>rgbcolor</code>	\triangleq	a synonym for <code>color</code>
<code>cmykcolor c</code>	\triangleq	true if <code>c</code> is a CMYK colour
<code>cyanpart(<i>c, y, m, k</i>)</code>	\triangleq	<code>c</code>
<code>magentapart(<i>c, y, m, k</i>)</code>	\triangleq	<code>y</code>
<code>yellowpart(<i>c, y, m, k</i>)</code>	\triangleq	<code>m</code>
<code>blackpart(<i>c, y, m, k</i>)</code>	\triangleq	<code>k</code>

Colour expressions usually appear in `withcolor` clauses in conjunction with commands such as `draw` (§5.1.3:29), `drawoptions` (§5.2.1:31), `fill` and `filldraw` (§5.1.5:30). `withcolor` works for both RGB and CMYK colours, but the two systems should not be mixed within a single picture.

3.4 Numerics

METAPOST uses fixed-point arithmetic with $2^{-16} = 1/65536$ as the unit for the `numeric` type. The largest number that can be represented is about 4096.

The lengths that METAPOST can represent range from $1/65536$ bp ≈ 5.38 nanometres to 4096 bp ≈ 1.4 metres. For comparison, the wavelength of blue light is about 400 nanometres and the length of standard paper is about 0.29 metres, depending on where you live.

The arithmetic operators `+`, `-`, `*`, and `/` can be used with numerics. The binary comparison operators `=`, `<>`, `<`, `<=`, `=>`, and `>` accept `numeric` operands and provide the conventional ordering.

Numeric constants and expressions include:

<code>epsilon</code>	\triangleq	1/65536 (the smallest value that METAPOST can represent)
<code>infinity</code>	\approx	4095.99998 (the largest value that METAPOST can represent)
<code>day</code>	\triangleq	current day (of the month)
<code>month</code>	\triangleq	current month
<code>year</code>	\triangleq	current year
<code>time</code>	\triangleq	job start time (minutes since midnight)

3 Types

Numeric functions include:

<code>decimal n</code>	\triangleq	the decimal string corresponding to n
<code>numeric n</code>	\triangleq	<code>true</code> if n is numeric
<code>odd n</code>	\triangleq	true if the closest integer to n is odd
<code>m div n</code>	\triangleq	$\lfloor m/n \rfloor$ (integer division)
<code>m mod n</code>	\triangleq	$m - n\lfloor m/n \rfloor$ (integer remainder)
<code>abs x</code>	\triangleq	$ x $, the absolute value of x
<code>sqrt x</code>	\triangleq	\sqrt{x}
<code>sind x</code>	\triangleq	$\sin x$ (x in degrees)
<code>cosd x</code>	\triangleq	$\cos x$ (x in degrees)
<code>mexp x</code>	\triangleq	$e^{x/256}$
<code>mlog x</code>	\triangleq	$256 \ln x$
<code>floor x</code>	\triangleq	the greatest integer less than or equal to x
<code>ceiling x</code>	\triangleq	the least integer greater than or equal to x
<code>round x</code>	\triangleq	the integer closest to x
<code>x++y</code>	\triangleq	$\sqrt{x^2 + y^2}$
<code>x+-y</code>	\triangleq	$\sqrt{x^2 - y^2}$
<code>uniformdeviate x</code>	\triangleq	a random number uniformly distributed in $[0, x]$
<code>normaldeviate</code>	\triangleq	a random number with normal distribution, $\mu = 0$, $\sigma = 1$

3.5 Pairs

A **pair** is a tuple of two **numerics** that is most often used to represent a position in the two-dimensional plane of the picture but can also be used to represent a direction. Literal pairs are written as `(m,n)` in which `m` and `n` are **numerics**.

Pairs can be added and subtracted (like vectors), multiplied or divided by **numerics**, and used in linear equations. *Mediation* abbreviates a common operation:


$$\mathbf{r}[\mathbf{p}, \mathbf{q}] \triangleq \mathbf{p} + \mathbf{r} * (\mathbf{q} - \mathbf{p}).$$

If $0 \leq \mathbf{r} \leq 1$, the result is a point between the points `p` and `q`, with $\mathbf{r} = 0$ corresponding to `p` and $\mathbf{r} = 1$ corresponding to `q`. All values of `r` are legal:

```

z1 = (0,0);
z2 = (0,50);
draw z1--z2;
dotlabel.rt(btex $4 \over 3$ etex, 4/3[z1,z2]);
dotlabel.rt(btex $2 \over 3$ etex, 2/3[z1,z2]);
dotlabel.rt(btex $1 \over 3$ etex, 1/3[z1,z2]);

```




3 Types

The names **x**, **y**, and **z** play a special role:

- They are undefined by `beginfig` (§2.2:2).
- $\mathbf{z}\langle\text{suffix}\rangle$ is defined to be equal to $(\mathbf{x}\langle\text{suffix}\rangle, \mathbf{y}\langle\text{suffix}\rangle)$, where $\langle\text{suffix}\rangle$ is a number, index, or name (§2.7:5). Thus `z[3].t` can be abbreviated to `z3t` and is equivalent to the pair `(x3t,y3t)`.

The pair `(x,y)` can be used to represent the direction $\theta = \tan^{-1}(y/x)$. Angles in METAPost are measured in degrees. For example, `dir 90` \equiv `pair(0,1)`.

```
smallRad = 30;
bigRad = 40;
for angle = 0 step 30 until 330:
  s := sind angle;
  c := cosd angle;
  drawarrow origin -- (smallRad*c, smallRad*s);
  label(decimal angle, (bigRad*c, bigRad*s));
endfor;
```



Constant pairs include:

<code>origin</code>	\triangleq	<code>(0,0)</code>
<code>up</code>	\triangleq	<code>(0,1)</code>
<code>down</code>	\triangleq	<code>(0,-1)</code>
<code>left</code>	\triangleq	<code>(-1,0)</code>
<code>right</code>	\triangleq	<code>(1,0)</code>

Functions for pairs include:

<code>xpart(x,y)</code>	\triangleq	<code>x</code>	
<code>ypart(x,y)</code>	\triangleq	<code>y</code>	
<code>abs z</code>	\triangleq	$\sqrt{x^2 + y^2}$	(hypotenuse)
<code>unitvector z</code>	\triangleq	<code>z/abs z</code>	(normalize)
<code>angle z</code>	\triangleq	$\tan^{-1}(y/x)$	(in degrees)
<code>round z</code>	\triangleq	<code>(round x, round y)</code>	
<code>z₁ dotprod z₂</code>	\triangleq	<code>x₁y₁ + x₂y₂</code>	(inner product)
<code>r[z₁, z₂]</code>	\triangleq	<code>z₁ + r(z₂ - z₁)</code>	(mediation)
<code>dir θ</code>	\triangleq	<code>(cosd θ, sind θ)</code>	(direction)

3.6 Pens

All objects are drawn with a pen. The command

```
pickup <pen expression>
```

3 Types

selects a pen with characteristics defined by $\langle pen\ expression \rangle$ with syntax

$$\left. \begin{array}{l} \text{pencircle} \\ \text{pensquare} \end{array} \right\} \langle transform \rangle$$

where $\langle transform \rangle$ is an affine transformation (§3.7:17). Naturally, **pencircle** gives a round nib and **pensquare** gives a square nib. The operator **pensquare** is actually a macro defined by

$$\text{pensquare} \triangleq \text{makepen}((-0.5,-0.5)--(0.5,-0.5)--(0.5,0.5)--(-0.5,0.5)--cycle)$$

and, in general, **makepen** will construct a pen nib from any closed path (§4:21). The operator **makepath** is the inverse of **makepen**: if p is a pen, then **makepath** p is the polygon that it uses as a nib. Thus:

$$\text{makepath pencircle} \equiv \text{fullcircle}.$$

It is best to use a round nib (**pencircle**) when drawing dotted and dashed lines (§4.2:22); square or polygonal pens yield unpredictable results.

The constant **nullpen** is a pen with no useful properties. A function that is supposed to return a pen but cannot do so should return **nullpen**.

The default pen is quite thin:


$$\text{defaultpen} \triangleq \text{pencircle scaled } 0.5 \text{ bp}.$$

Thick pens are defined by simple scaling:

$$\text{pencircle scaled } 5\text{pt}$$

You can use **xscaled** and **yscaled** to obtain an elliptical or rectangular nib:

```
path p;
drawoptions(withpen pencircle
  xscaled 2bp yscaled 0.1bp);
p = (0,0) for i = 1 upto 100:
  .. (0.6i*cosd(36i),0.6i*sind(36i)) endfor;
draw p;
drawoptions();
```



68

The shape of the end of each line is determined by the internal variable **linecap**. When lines change direction, the shape of the corner is determined by **linejoin**. Figure 1 illustrates both.

$$\text{linecap} := \left\{ \begin{array}{ll} \text{rounded} & (\text{default}) \\ \text{squared} & \\ \text{butt} & \end{array} \right. \quad \text{linejoin} := \left\{ \begin{array}{ll} \text{rounded} & (\text{default}) \\ \text{beveled} & \\ \text{mitered} & \end{array} \right.$$

A long spike may be produced when **mitered** lines meet at an acute angle. METAPost changes the join shape to **beveled** if

$$\frac{\text{miter length}}{\text{line width}} > \text{miterlimit}.$$




Figure 1: The effect of linecap and linejoin, drawn withpen pensquare scaled 8

The default value of the internal variable `miterlimit` is 10.0; it can be changed by assignment. For a sequence of `draw` (§5.1.3:29) and `fill` (§5.1.5:30) commands with a particular kind of pen, use


```
drawoptions ( <text> )
```

to set the pen characteristics and

```
drawoptions ( )
```

to restore the defaults. The argument `<text>` can specify `dashed` (§4.2:22), `withcolor` (§3.3:11), and `withpen` (§3.6:14) values.

```
drawoptions(
  withcolor 0.8 green
  withpen pensquare scaled 4 );
d = 10;
draw unitsquare scaled 2d shifted (-d,-d);
draw fullcircle scaled 4d;
drawoptions();
```



26


3.7 Transforms

A variety of affine transformations can be applied to pictures. These are transform expressions (all angles are measured in degrees):

(x, y)	<code>rotated</code>	θ	\triangleq	$(x \cos \theta - y \sin \theta, x \sin \theta + y \cos \theta)$
(x, y)	<code>slanted</code>	a	\triangleq	$(x + ay, y)$
(x, y)	<code>scaled</code>	a	\triangleq	(ax, ay)
(x, y)	<code>xscaled</code>	a	\triangleq	(ax, y)
(x, y)	<code>yscaled</code>	a	\triangleq	(x, ay)
(x, y)	<code>shifted</code>	(a, b)	\triangleq	$(x + a, y + b)$
(x, y)	<code>zscaled</code>	(a, b)	\triangleq	$(ax - by, bx + ay)$
(x, y)	<code>reflectedabout</code>	(p, q)	\triangleq	The reflection of (x, y) in the line (p, q) .
(x, y)	<code>rotatedaround</code>	(p, θ)	\triangleq	(x, y) rotated about point p through angle θ .

Transforms are applied by writing a transform expression after a picture expression. Transform expressions can be combined by concatenating them.

```
dotlabel.bot(btex (0,0) etex, origin);
path bx;
bx = unitsquare scaled 10;
draw bx shifted (0,36);
draw bx xscaled 1.6 shifted (0,24);
draw bx slanted 0.5 shifted (0,12);
draw bx rotated 30;
```



As usual, the type may be used as a predicate:

`transform t` \triangleq true if t is a transform.

Variables of type `transform` may be declared. If t is a transform variable, then

`p transformed t`

is the picture p transformed by t and

`inverse t`

is the inverse of t . The constant `identity` is a transform that has no effect:

`draw p transformed identity` \equiv `draw p`

The following example defines and applies a transform called `reflect`.

```
transform reflect;
reflect = identity xscaled -1;
draw btx Mirror, mirror, on the wall etex
transformed reflect;
```

Mirror on the wall

27

3 Types

A transform has six parameters. The mapping from (x, y) to (x', y') is given, in homogeneous coordinates, by


$$\begin{bmatrix} x' \\ y' \\ 1 \end{bmatrix} = \begin{bmatrix} t_{xx} & t_{xy} & t_x \\ t_{yx} & t_{yy} & t_y \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} x \\ y \\ 1 \end{bmatrix}$$

The values of the parameters for a transform T are referred to in METAPOST as

$$\begin{array}{ll} t_x &= \text{xpart } T \\ t_{xx} &= \text{xxpart } T \\ t_{xy} &= \text{xypart } T \end{array} \quad \begin{array}{ll} t_y &= \text{ypart } T \\ t_{yx} &= \text{yxpart } T \\ t_{yy} &= \text{yypart } T \end{array}$$

Transforms may be defined by equations. Each equation says that a point p is mapped to a point p' and determines two of the transform parameters; consequently, three equations are needed. In the following example, the transform t is defined by its effect on three points.

```
z1 = (-20,0); z2 = (-40,15);
z3 = (0,20); z4 = (0,30);
z5 = (20,0); z6 = (35,-20);
draw fullcircle scaled 40;
drawarrow z1--z2;
drawarrow z3--z4;
drawarrow z5--z6;
transform t;
z1 transformed t = z2;
z3 transformed t = z4;
z5 transformed t = z6;
draw fullcircle scaled 40 transformed t;
```



28


The parameters themselves can be used to define the transformation. For example, the equations

```
xxpart t = yypart t;
xypart t = -yxpart t;
```

specify that transform t preserves shapes. Two additional equations, defining the effect of t on two points, are required to define t completely.

The following example combines several features of METAPOST: the command **direction** (§4.5:24) gives the direction of the tangent vector at a point on the curve; the command **angle** (§3.4:12) converts this direction to an angle θ ; rotating by $\theta + 90^\circ$ gives a direction normal to the curve; and the command **point** (§4.5:24) gives the position of a point on the curve.

```
path p;
p := fullcircle xscaled 100 yscaled 40;
draw p;
for i = 0 upto 40:
  t := i/5;
  draw ((0,0)--(3,0))
    rotated (angle direction t of p - 90)
    shifted (point t of p);
endfor;
```



47

3.8 Pictures

Anything that can be drawn by METAPOST can be stored in a picture variable. Several primitives, including `draw` (§4:21), store their results in the internal variable `currentpicture`. Pictures can be transformed (§3.7:17).

The following expressions yield a picture:

```

s infont f : string s set in font f (§3.2:10)
btex ... etex : raw TEX strings (§5.1.1:28)
TEX( ... ) : processed TEX strings (§7.3 : 43)
nullpicture : an empty picture
currentpicture : the “current” picture, destination of draw, etc.

```

These operators return a value associated with a picture p :

```

center p  $\triangleq$  the centre of  $p$ .
length p  $\triangleq$  the number of components of  $p$ .
llcorner p  $\triangleq$  the lower left corner of  $p$ .
lrcorner p  $\triangleq$  the lower right corner of  $p$ .
ulcorner p  $\triangleq$  the upper left corner of  $p$ .
urcorner p  $\triangleq$  the upper right corner of  $p$ .

```

After defining the picture `p` with the commands

```

picture p;
p = btex $\displaystyle\int_0^\infty e^{-x} \sin x \, dx$ etex;

```

we can draw the picture and its corner points:

```

draw p;
dotlabel.ulft("UL", ulcorner p);
dotlabel.ur ("UR", urcorner p);
dotlabel.llft("LL", llcorner p);
dotlabel.lrt ("LR", lrcorner p);

```

$$\begin{array}{ccc} \text{UL} & & \text{UR} \\ \bullet & \int_0^\infty e^{-x} \sin x \, dx & \bullet \\ \text{LL} & & \text{LR} \end{array} \quad \# 2$$

The operators `llcorner`, `lrcorner`, `ulcorner`, and `urcorner` define the *bounding box* of a picture. They can be used to measure the size of a box. Following on from the above example,

```

pair dim;
dim := (urcorner p) - (llcorner p);
width := xpart dim;
height = ypart dim;

```

gives `width` = 71.3919 and `height` = 23.1722.


If the bounding box is not what you want, you can change it. The command

3 Types

`setbounds v to p`

makes the picture variable v behave as if its bounding box is the path p .

```
picture p;
p = btex $\displaystyle\sum_{n=1}^{\infty} \{1 \over n\}$ etex;
path q;
q = bbox p shifted (20,20);
setbounds p to q;
draw p;
draw q;
```



67

Recent versions of METAPost include a macro `image` that can be used in places that would otherwise require `currentpicture`. The expression

`image($\langle seq \rangle$)`

yields the picture object constructed by the sequence $\langle seq \rangle$ of drawing commands.

```
picture smiley;
smiley := image(
  draw fullcircle scaled 40;
  draw halfcircle scaled 20 rotated 180;
  filldraw fullcircle scaled 3 shifted (-10,10);
  filldraw fullcircle scaled 3 shifted (10,10);
);
draw smiley;
```



70

The following predicates can be applied to any object, but are most meaningful for parts of a picture. A “stroked picture component” is a part of a picture that was drawn by moving a pen (as opposed to setting text, etc.). These predicates can be used to analyze the structure of a picture (§5.2.3:32).

<code>bounded x</code>	\triangleq	x is a picture with a bounding box
<code>clipped x</code>	\triangleq	x is a picture that has been clipped
<code>dashpart x</code>	\triangleq	x is the dash pattern of a path in a stroked picture
<code>filled x</code>	\triangleq	x is a filled outline
<code>pathpart x</code>	\triangleq	x is the path of a stroked picture component
<code>penpart x</code>	\triangleq	x is the pen of a stroked picture component
<code>stroked x</code>	\triangleq	x is a stroked line
<code>textual x</code>	\triangleq	x is typeset text

3.9 Lists

Lists have a rather ghostly existence in METAPost: there is no type *list*, but there are a few constructions that create and consume lists.

4 Paths

The function `scantokens` converts a string to a list of tokens. The expression `x step d until y` generates the list $x, x + d, x + 2d, \dots$, stopping when $x + nd \geq y$. For the common case $d = 1$, there are macros `upto` and `downto`:

$$\begin{aligned} \text{upto} &\triangleq \text{step } 1 \text{ until} \\ \text{and} \quad \text{downto} &\triangleq \text{step } -1 \text{ until} \end{aligned}$$

The `for` statement (§5.2.3:32) has a form

```
for <var> = <list>: <seq> endfor
```

which assigns the variable `<var>` to each item in `<list>`. The list can be an actual list of tokens separated by commas, as in

```
for i = 1, 2, 4, 8, 16: ... endfor
```

or a list generated by `scantokens`, as in

```
for i = scantokens("1, 2, 4, 8, 16"): ... endfor
```

4 Paths

Paths are piecewise straight lines, curves, and any combination of these. Path variables of type `path` may be declared and there are various kinds of *path expressions*. The following examples use the command `draw` (§5.1.3:29) to draw paths.

4.1 Straight lines

The binary operator `--` constructs a straight line between its point operands. It may be used in a sequence, so that

```
draw z0 -- z1 -- z2 -- z3;
```

draws three line segments, connecting `z0` to `z1`, and so on.

If the last point is the internal name `cycle`, the last line segment returns to the starting point. The keyword `cycle` can also be used as a predicate:

$$\text{cycle } p \triangleq \text{true if path } p \text{ is a cycle}$$

The expression `p & q` yields the concatenation of the paths `p` and `q`, provided that the last point of `p` is identical to the first point of `q`.

4.2 Dots and Dashes

The general syntax for a dashed line is:

```
draw <path> dashed <pattern>
```


The basic patterns are:

```
evenly : evenly-spaced dashes, 3 bp long,
and withdots : dots 5 bp apart.
```

Patterns may be transformed (§3.7:17):

```
shifted <pair> : each dash is displaced by <pair>,
and scaled <numeric> : each dash is scaled by <numeric>.
```

```
draw (0,30)--(120,30) dashed withdots;
draw (0,20)--(120,20) dashed evenly;
draw (0,10)--(120,10) dashed evenly shifted (3,3);
draw (0, 0)--(120, 0) dashed evenly scaled 1.5;
```



3

Dashed and dotted lines are best drawn using a pen with a circular nib (§3.6:14).

4.3 Circles, Disks, and Arcs

The command `fullcircle` draws a circle with unit diameter at the origin. Apply transformations (§3.7:17) to the circle to modify the size and position. The command `halfcircle` is similar to `fullcircle` but draws only the part of the circle above the X axis. The command `quartercircle` draws only the first quadrant of the circle.

```
draw fullcircle scaled 50;
draw halfcircle scaled 40;
draw quartercircle scaled 30;
filldraw fullcircle scaled 20 withcolor 0.3white;
```



General arcs can be drawn using the `subpath` command (§4.5:24). The path length of a full circle is 8, and so one unit corresponds to 45° . Thus the arcs in the picture below have lengths of 135° , 225° , and 315° .

```
draw subpath (0, 3) of fullcircle scaled 20;
draw subpath (0, 5) of fullcircle scaled 30;
draw subpath (0, 7) of fullcircle scaled 40;
```



4.4 Curves


The binary operator ‘`..`’ is used like ‘`--`’, but it draws cubic Bézier curve instead of straight lines. The command

```
draw z0 .. z1 .. z2 .. z3;
```

draws a Bézier curve that passes through the points `z0`, `z1`, `z2`, and `z3`.

The binary operator ‘`...`’ is similar to ‘`..`’ but tries to avoid inflection points. The difference between ‘`..`’ and ‘`...`’ is generally small, but can be seen in the following example, in which the lower curve uses ‘`...`’.


```
z0 = (0,0); z1 = (50,25); z2 = (150,50);
draw z0{dir -20} .. z1 .. {dir 30}z2;
draw (z0{dir -20} ... z1 ... {dir 30}z2) shifted (0,-10);
```



53

The binary operator ‘`---`’ is similar to ‘`--`’ but ensures a smooth transition between straight and curved sections of a path, as illustrated in the two examples below.

```
draw (0,0) -- (25,0) .. (50,25);
```



51


```
draw (0,0) --- (25,0) .. (50,25);
```



52

The direction of the curve may be specified at a point by a qualifier `{p}`, in which `p` is a pair representing a direction (§3.5:13). The qualifier may be placed either before or after the point.


```
for i = 0 step 30 until 120:
  draw (0,0){dir i} .. (100,0);
endfor;
```



12

The curvature of the curve can be specified at its end points using `curl`. The argument of `curl` can have any positive value; smaller values give smaller curvature.


```
draw (0,0){curl 1} .. (50,30) .. {curl 1}(100,0);
draw (0,0){curl 5} .. (50,-30) .. {curl 5}(100,0);
```



44

The **tension** of a curve may be changed between any pair of points. The minimum value of **tension** is $\frac{3}{4}$; the default value is 1; higher values give straighter curves.

```
for i = 1 upto 5:
  draw (0,0){dir 90} .. tension i .. {dir 270}(100,0);
endfor;
```



Control points are points that are not on the curve but which “attract” the curve. They can be specified between points with syntax “**controls** *p* and *q*”, where *p* and *q* are points.

```
z1 = (10,25);
z2 = (90,25);
dotlabel("", z1);
dotlabel("", z2);
draw (0,0) .. controls z1 and z2 .. (100,0);
```



4.5 Parametric Paths

A path may be considered as a set of points $\{(X(t), Y(t))\}$, where t is the path *parameter*. Given a value of t , we can find the corresponding point $(X(t), Y(t))$ on the path and the direction of the path. Conversely, given a point or a direction on the path, we can find the value of t . We can use t to find where curves intersect and to extract parts of curves.




Figure 2: Parametric curves

We illustrate concepts related to parametric curves using Figure 2. First we define four key points and draw the path, *a*, through them.

```
z1 = (0,0);
z2 = (75,75);
z3 = (150,0);
z4 = (225,75);
path a;
a = z1 .. z2 .. z3 .. z4;
draw a;
```

The path has a *length* associated with it:

4 Paths

```
show length a; >> 3
```

It is convenient to think of the path being drawn by a point which moves along it, starting at time 0 and ending at time 3. The “times” of the control points are 0 for **z1**, 1, for **z2**, 2 for **z3**, and 3 for **z2**. The expression `point...of` finds the point on a path corresponding to a particular time (note the position of **z5** in the diagram):

```
z5 = point 2.5 of a;
```

We can find the direction of the path, expressed as a tangent vector, at any time.

```
show direction 0 of a; >> (-19.5705,47.24738)
```

If we prefer, we can work with the arclength instead of the time:

```
lena = arclength a;
show lena; >> 378.67822
t1 = arctime lena/5 of a;
show t1; >> 0.56165
z6 = point t1 of a;
```

The expression `subpath (t_1, t_2) of p` returns the part of the path p between times t_1 and t_2 . The effect of

```
draw subpath (0,t1) of a shifted (-10,0);
```

can be seen in Figure 2 as the short arc on the left.

We can find the time on a path when the direction first achieves a particular value:

```
t2 = directiontime (1,-1) of a;
z7 = point t2 of a; >> 1.21132
show t2;
```

The straight line in Figure 2 is drawn as path **b** with the commands:

```
z8 = (25,0);
z9 = (200,75);
path b;
b = z8 -- z9;
draw b;
```

The point of intersection of paths **a** and **b** is given by:

```
z10 = a intersectionpoint b;
```

and the command

```
show a intersectiontimes b; >> (1.5,0.5)
```

shows that the intersection occurs at time 1.5 for path **a** and at time 0.5 for path **b**. If the paths **a** and **b** do not intersect, the result is $(-1,-1)$.

The expression

4 Paths

`a cutbefore b`


yields the part of path `a` from `z1` to `z10` and the expression

`a cutafter b`

yields the part of path `a` from `z10` to `z4`.

A circle drawn by METAPOST has eight control points:

```
diam =100;
rad = 0.5*diam+8;
path circ;
circ = fullcircle scaled diam;
draw circ;
for i = 0 upto 7:
  dotlabel("", point i of circ);
  ang := 45 * i;
  label(decimal i, (rad*cosd(ang), rad*sind(ang)));
endfor;
```



77

Summary of path expressions:

Path expressions yielding **boolean**:

`path p` \triangleq p is a path
`cycle p` \triangleq p is a cycle (closed path)

Path expressions yielding **numerics**:

`length a` \triangleq the value of t at the end of a
`arctime ℓ of a` \triangleq the value of t at which the length of a is ℓ
`directiontime d of a` \triangleq the value of t at which a first has the direction d
`arclength a` \triangleq the total arc length of a

Path expressions yielding **pairs**:

`point t of a` \triangleq the position of a at time t
`direction t of a` \triangleq the direction of a at time t
`directionpoint d of a` \triangleq the first point at which the direction of a is d
 `a intersectionpoint b` \triangleq the point at which a intersects b
 `a intersectiontimes b` \triangleq the pair (t_a, t_b) of intersection times

Path expressions yielding **paths**:

`fullcircle` \triangleq the unit circle (origin at centre)
`unitsquare` \triangleq the unit square (size 1×1 , origin at lower left corner)
`bbox a` \triangleq the bounding box of path a
`reverse a` \triangleq the path a drawn backwards

$\text{subpath}(t_1, t_2) \text{ of } a \triangleq$ the part of a for which $t_1 \leq t \leq t_2$
 $a \& b \triangleq$ the concatenation of paths a and b
 $a \text{ cutbefore } b \triangleq$ the part of a before its intersection with b
 $a \text{ cutafter } b \triangleq$ the part of a after its intersection with b

4.6 Path Constructors

The following functions and macros construct paths:

$\text{reverse } p \triangleq$ the path p with its direction reversed
 $\text{unitsquare} \triangleq (0,0) \text{ --}(1,0) \text{ --}(1,1) \text{ --}(0,1) \text{ -- cycle}$
 $\text{bbox } p \triangleq \text{llcorner } p \text{ -- lrcorner } p \text{ -- urcorner } p \text{ -- ulcorner } p \text{ -- cycle}$

The effect of `reverse` is defined by

$\text{point } t \text{ of reverse } p \equiv \text{point}(\text{length } p - t) \text{ of } p$

The `unitsquare` can be transformed (§3.7:17) to produce an arbitrary parallelogram. The macro `bbox`, applied to a picture p , returns the bounding box of p .

```

picture p;
p = btex $\spadesuit \heartsuit \diamondsuit \clubsuit$ etex;
draw p;
draw bbox p;

```




19

The command `buildcycle` is especially useful for boundaries of shaded areas (§5.1.5:30). Given several paths, `buildcycle` tries to piece them together to form an enclosed figure, which is returned as the result of the expression. Draw the bounding path, if required, *after* shading the interior area with `filldraw`.

```

path p[];
p1 = (10,-5) -- (10,50);
p2 = (-5,0) -- (90,0);
p3 = (80,-5) -- (80,10);
p4 = (10,40) for i = 2 upto 9:
    .. (10*i, 40/i) endfor;
p5 = buildcycle(p1, p2, p3, p4);
filldraw p5 withcolor 0.7 white;
draw p1; draw p2; draw p3; draw p4;

```



41

Font	Name	Sizes: $s =$
Bold	<code>cmb$\langle s \rangle$</code>	10
Bold extended	<code>cmbx$\langle s \rangle$</code>	5, 6, 7, 8, 9, 10, 12
Italic	<code>cmti$\langle s \rangle$</code>	7, 8, 9, 10, 12
Roman	<code>cmr$\langle s \rangle$</code>	5, 6, 7, 8, 9, 10, 12, 17
Sans serif	<code>cmss$\langle s \rangle$</code>	8, 9, 10, 12, 17
Sans serif italic	<code>cmssi$\langle s \rangle$</code>	8, 9, 10, 12, 17
Slanted	<code>cmsl$\langle s \rangle$</code>	8, 9, 10, 12
Small caps	<code>cmcsc$\langle s \rangle$</code>	10
Typewriter	<code>cmtt$\langle s \rangle$</code>	8, 9, 10, 12
Typewriter italic	<code>cmitt$\langle s \rangle$</code>	10
Typewriter slanted	<code>cmslitt$\langle s \rangle$</code>	10

Figure 3: L^AT_EX fonts

5 Commands

5.1 Drawing Commands

5.1.1 `btex` and `verbatimtex`

The command `btex $\langle text \rangle$ etex` invokes T_EX to process $\langle text \rangle$ and returns the result as a picture:

```
draw btex  $\sqrt{x^2+y^2}$  etex;  $\sqrt{x^2+y^2}$  # 17
```

Commands between `btex` and `etex` are written in T_EX, *not* L^AT_EX, and the text is set in HR mode. For example, `$$... $$` does not work, but `$\displaystyle ...$` usually gives the right effect. For more elaborate effects, use the T_EX package (§7.3:43).

Text is set in the default font, which is usually `cmr10`. The default font can be changed by assigning to the internal variable `defaultfont`. The size of the text is determined by the default scale, which is 1. The scale can be changed by assigning to the variable `defaultscale`.

The command `verbatimtex $\langle text \rangle$ etex` is similar, but does not produce a picture. It is used to configure T_EX for subsequent processing. For example, the following commands define the font commands `\bkm` to be 11pt Bookman and `\lit` to be 10pt Typewriter:

```
verbatimtex
  \font\bkm = pbkli scaled 1100
  \font\lit = cmtex10 scaled 1000
etex
```

These fonts can then be used in a `btex` command:

```
label.bot(btex\lit server etex, 0.5[a.e, b.w]);
```

Figure 3 shows some of the standard L^AT_EX fonts.

5 Commands

The command `verbatimtex` can be used to create an environment in which \LaTeX can be used in `btex` commands. The following METAPost file creates a \LaTeX table:

```
verbatimtex
%&latex
\documentclass{article}
\begin{document}
etex

beginfig(1);
label(btex \begin{tabular}{c} First\\Second \end{tabular} etex, origin);
endfig;

end
```


5.1.2 clip

The command

`clip p to b`

draws all parts of the picture p that lie within the boundary b . The picture p must be given as a picture variable, not a picture expression. The argument b is a path.

```
for i = -50 step 5 until 50:
  draw (i,-50) -- (i+100,50);
endfor;
path c;
c = fullcircle scaled 70 shifted (50,0);
clip currentpicture to c;
draw c;
```



54

5.1.3 draw

The command

`draw p`

draws the path or picture defined by the picture expression p . A picture can be erased by drawing it with the background colour:

$$\text{undraw } p \triangleq \text{draw } p \text{ withcolor background}$$

5.1.4 drawarrow

The following commands draw paths with arrows. The argument of each of these commands is a path, not a picture.

```
drawarrow  $p$  : draw path  $p$  with an arrow at the end
drawarrow reverse  $p$  : draw path  $p$  with an arrow at the start
drawdblarrow  $p$  : draw path  $p$  with an arrow at both ends
```

The size of the arrow head is determined by `ahlength` and `ahangle`. These internal variables have default values `ahlength = 4` and `ahangle = 45°`, respectively. The code in the following example draws an arrow with default values and another arrow with length 10 and angle 30°.

```
begingroup;
  drawarrow (0,30) -- (60,30);
  interim ahlength := 10;
  interim ahangle := 30;
  drawarrow reverse ((0,0) -- (60,0));
endgroup;
```



5.1.5 fill

The following commands “paint” areas of the figure. The path must be a cycle (§4.1:21), otherwise the paint may leak out.


```
fill  $p$  : paint the area enclosed by path  $p$  with black
fill  $p$  withcolor  $c$  : paint the area enclosed by path  $p$  with colour  $c$ 
```

There are some useful macros that use `fill`:

```
filldraw  $p$   $\triangleq$  draw (fill  $p$ )
filldraw  $p$  withcolor  $c$   $\triangleq$  draw (fill  $p$  withcolor  $c$ )
unfill  $p$   $\triangleq$  fill  $p$  withcolor background
```

Filling can be used to avoid tricky coding with `cutbefore` and `cutafter`. In this example, `filldraw` erases the part of the line that would be inside the circle at `z2`.

```
z1 = (0,20); z2 = (0,-20);
draw z1--z2;
filldraw fullcircle scaled 20 shifted z2 withcolor white;
draw fullcircle scaled 20 shifted z1;
draw fullcircle scaled 20 shifted z2;
```



5.1.6 label

The command

```
label <suffix> ( <picture expression> , <pair> )
```


writes a label. $\langle picture \rangle$ describes the label to be written. It is usually a string (§3.2:10) or `btex ... etex` (§5.1.1:28). The $\langle pair \rangle$ determines the position of the label. The $\langle suffix \rangle$ may be omitted; if it is present, it must be one of

```
.lft .rt .top .bot .ulft .urt .llft .lrt
```

and it defines the position of the label with respect to $\langle pair \rangle$.


The command `dotlabel` is similar to `label` but draws a large dot at the labelled point.

```
draw (0,0) -- (30,20) -- (-30,20) -- cycle;
label.ulft(btex (-30,20) etex, (-30,20));
label.urtr(btex (30,20) etex, (30,20));
dotlabel.bot(btex (0,0) etex, (0,0));
```



The command `dotlabels` provides an abbreviation that can be used when all of the points to be labelled have the form $z\langle suffix \rangle$ and the suffixes are suitable for use as labels.

```
z.alpha = (0,0);
z.beta = (20,20);
z.gamma = (40,0);
dotlabels.top(alpha, beta, gamma);
```



The command `thelabel` is also similar to `label` but returns the result as a picture without drawing it. Thus:

$$\text{label}(p, z) \triangleq \text{draw thelabel}(p, z)$$

5.2 Non-drawing Commands

5.2.1 drawoptions

The command `drawoptions(<text>)` adds the options specified by $\langle text \rangle$ to all drawing commands until it is cancelled by the command `drawoption()`. The options allowed in $\langle text \rangle$ include `dashed` (§4.2:22), `withcolor` (§3.3:11), and `withpen` (§3.6:14). For example:

```
drawoptions(withcolor blue);
drawboxed(fred);
drawoptions();
```

The options in the scope of `drawoptions` affect only the commands for which they “make sense”. For example, if a picture is drawn with `drawoptions(dashed evenly)`, all of its paths will be drawn with dashes, but its labels will not be affected.

5.2.2 `filenametemplate`

By default, METAPOST reads from a file called `<jobname>.mp` and writes the output created by `beginfig(n) ... endfig` to `jobname.n`. The name of the output file can be changed by including a command of this form near the beginning of the input file:

```
filenametemplate <format string> ;
```

The resulting file name is the `<format string>` after escape sequences have been processed. Figure 4 lists the escape codes that METAPOST recognizes.

<code>%%</code>	Percent sign
<code>%j</code>	Job name
<code>%δc</code>	The argument of <code>beginfig</code>
<code>%δy</code>	Year
<code>%δm</code>	Month
<code>%δd</code>	Day
<code>%δH</code>	Hour
<code>%δM</code>	Minute

Figure 4: Escape sequences for `filenametemplate`. The symbol δ stands for an optional decimal digit (0, 1, 2, ..., 9) that sets the size of the string.

Suppose that the input file is `doc.mp`. By default, METAPOST behaves as if

```
filenametemplate "%j.%c";
```

has been executed and generates figure files called `doc.1`, `doc.2`, etc. If the METAPOST program included the command

```
filenametemplate "doc-%4y-%3c.mps";
```

then the output file names would be `doc-2007-001.mps`, `doc-2007-002.mps`, etc. This is useful for version of L^AT_EX that do not recognize numbers as file extensions.

5.2.3 `for`

The METAPOST loop has the general form:

```
for <name> = <list>:
  <seq>
endfor
```

Although *list* is not a type of METAPOST, there are expressions that generate lists (§3.9:20). For example:

```
for n = 1 upto 10:
  show x;
endfor
```

in which `upto` is an abbreviation, as is `downto`:

$$\begin{aligned} \text{upto} &\triangleq \text{step } 1 \text{ until} \\ \text{and} \quad \text{downto} &\triangleq \text{step } -1 \text{ until} \end{aligned}$$

The numerical values do not have to be integers:

```

                                >> 0
                                >> 0.1
for x = 0.5 step 0.1 until 1.5:
    show x;                      >> 0.20001
endfor;                          >> 0.30002
                                ....
                                >> 0.90005

```

A list of expressions is accepted:

```

for e = 1, 256.1, "done":       >> 1
    show e;                     >> 256.1
endfor;                         >> "done"

```

There are not many METAPost expressions that yield lists of tokens: one of them is `scantokens` (§3.9:20). Consequently, `for` and `scantokens` are often used together (§9.6:52).

If there are no natural loop variables, use

```
forever: <seq> exitif <pred>; <seq> endfor
```

or

```
forever: <seq> exitunless <pred>; <seq> endfor
```

A `for` loop can be used to obtain the parts of a picture. The effect of

```
for c within <picture expression> : <loop> endfor
```

is to bind each component of the picture in turn to *c* and then execute the text in *<loop>*. Various predicates and selectors may be used in *<loop>*, including `stroked`, `filled`, `textual`, `clipped`, `bounded`, `pathpart`, `penpart`, and `dashpart`. Figure 5 provides a simple example.

Like its relatives `TEX` and `metafont`, METAPost is more like a macroprocessor than a formal language. This means that it is often possible to arbitrary chunks of text in a `for` loop, not just complete expressions. The shading example on page 27 illustrates this kind of usage.

5.2.4 if

Tests can be used in conditional statements which have the general form

```

if <pred>: <seq>
else: <seq>
fi

```

Compound conditional statements use `elseif`:

```
draw (0,0) -- (100,0);
filldraw fullcircle scaled 50;
label.bot("line", (50,0));
picture p;
p = currentpicture;
show "p has " & decimal(length(p))
    & "components.";

n = 0;
string msg;
for i within p:
    n := n + 1;
    msg := "Component " &
        decimal n & " is";
    if stroked i:
        msg := msg & " stroked";
    fi;
    if filled i:
        msg := msg & " filled";
    fi;
    if textual i:
        msg := msg & " textual";
    fi;
    show msg;
endfor;
```

```
>> "p has 3 components."
>> "Component 1 is stroked"
>> "Component 2 is filled"
>> "Component 3 is textual"
```

Figure 5: Using `for...within` to obtain a partial analysis of a picture.

```
if <pred>: <seq>
elseif <pred>: <seq>
else <seq>
fi
```

For example (§7.1:39):

```
if pos = "top":
    label.top(title, name.n);
elseif pos = "bottom":
    label.bot(title, name.s);
fi;
```

5.2.5 message

The command `message` takes a `string` expression as its argument and displays the string on the console.

5.2.6 readfrom

The effect of

```
readfrom  $\langle file\ name \rangle$ 
```

is to read one line from the named file and return the line as a string. If the file cannot be read, or if end of file has been reached, the string returned is `EOF`, which contains just the null character (i.e., `EOF = "\0"`).

If `readfrom $\langle file\ name \rangle$` is executed again after it has returned `EOF`, the file is read again from the beginning.

See (§9.6:52) for an example of the use of `readfrom`. It is also possible to write to a file (§5.2.9:36).

5.2.7 save

The command `save`, followed by a list of variable names, saves the values of these variables and restores them again at the end of the scope, effectively making these variables local to the current scope. See also (§2.10:6) and (§6:36).

5.2.8 show and friends

The command

```
show  $e$ 
```

writes `>>` followed by the value of the expression e to standard output. If the expression cannot be expressed in a simple textual form, METAPOST writes the type instead.

```
show (0,0) -- (10,0);           >> path
```

There are variants of `show` that are used less often but can be handy for debugging:

<code>showdependencies</code>	\triangleq	the known dependencies for a set of equations
<code>showtoken</code>	\triangleq	the parameters and replacement text of a macro
<code>showvariable</code>	\triangleq	the properties associated with a name

The variable `showstopping` controls the behaviour of METAPOST after it has executed a `show` command. If `showstopping > 0`, METAPOST pauses after each `show` command.

5.2.9 write

The effect of the command

```
write <string expression> to <file name>
```

is to write one line to the named file. If the file is not already open, METAPost opens it before writing. The file is closed when the METAPost program terminates, or explicitly by executing

```
write EOF to <file name>
```

It is also possible to read from a file: (§5.2.6:35).

6 Macros

METAPost provides many ways of defining macros. Here we describe just the most basic and useful techniques.

6.1 def macros

Simple macros have the form

```
def <name> = <seq> enddef
```

Simple macros with parameters have the form

```
def <name> ( <parameters> ) = <seq> enddef
```

Following the macro definition, each occurrence of *<name>* is replaced by *<seq>*, with parameters replaced by arguments when applicable. If the sequence *<seq>* ends with an expression, the value of that expression becomes the value of the macro. Note that there is no semicolon after the expression:

```
def hyp(expr a, b) =
  sqrt(a*a + b*b)
enddef;
```

```
show hyp(3,4);
```

```
>> 5
```


We can obtain a simple three-dimensional effect by adding a shadow to an outline. If the light source is above and to the left, the shadow should be below and to the right. The macro **shadow** gives a shadow to any path that can be filled — that is, to any closed path.

```

def shadow(expr p) =
  filldraw p shifted (3,-3) withcolor 0.7 white;
  filldraw p withcolor white;
  draw p;
enddef;

shadow(unitsquare scaled 20);
shadow(fullcircle scaled 20 shifted (40,0));
path p;
p = (0,0)--(0,20)--(20,40)--(50,40)--
    (30,20)--(30,0)--cycle;
shadow(p shifted (70,-20));

```



46

There are three kinds of macro parameter, distinguished by writing one of the keywords `expr`, `suffix`, or `text` before the parameter list.

- An `expr` parameter can be used as a variable in the body of the macro. Its value can be constrained by equations but cannot be re-assigned. The following macro says that `a`, `b`, `c`, and `d` are symmetrically placed around the origin but their separation is not known until `z1` is defined:

```

def vx(expr a, b, c, d) =
  a + c = (0,0);
  b + d = (0,0);
  xpart a = xpart d;
  ypart a = ypart b;
enddef;

vx(z1,z2,z3,z4);
z1 = (1,1);
show z1; show z2; show z3; show z4;

```

>> (1,1)
 >> (-1,1)
 >> (-1,-1)
 >> (1,-1)

- A `suffix` parameter is bound to an initializing expression but may also be used as a variable name. The macro `showfirst` would be illegal with an `expr` parameter, but is allowed with a `suffix` parameter:

```
def showfirst(suffix a) = show a[0]; enddef;
```

- A `text` parameter is bound to an arbitrary sequence of tokens. When its name is encountered in the body, METAPOST simply substitutes the corresponding argument and parses the result. For example, if we define

```
def doIt(text cmd) = cmd (0,0)--(5,5); enddef;
```

then `doIt(draw)` draws the path but `doIt(show)` simply displays it.

If a macro has two or more different kinds of parameters, there must be a separate parameter list for each kind:

```
def tricky(expr a, b)(suffix p) = ... enddef;
```

However, the corresponding invocation has just a single list of arguments:

```
tricky(2.3, xpart q, z);
```


6.2 vardef and other macro forms

A macro definition may be introduced by `vardef` instead of `def`. The most useful difference is that the body of a `vardef` macro is a group. A macro such as `area` can be used in the same way as a function in other languages:

```
vardef area(expr a, b, c) =
  save s;
  s := 0.5*(a+b+c);
  sqrt(s*(s-a)*(s-b)*(s-c))
enddef;

show area(3,4,5);
```

>> 6

The name of a `vardef` macro may be a compound containing suffixes. Within the macro, `@` is the last token of the macro call and `#@` is everything that comes before the last token. Macros defined with `vardef` can be used to define special behaviour for particular variable names. For example, the special relationship between `x`, `y`, and `z` is a consequence of the macro definition

```
vardef z@# = (x@#, y@#) enddef;
```

There are other forms of macro definition for which a full description goes beyond the scope of this manual. For example, macros can be used to define unary and binary operators. After defining

```
vardef neg primary x = -x enddef;
```

we can use `neg` to negate numeric values and pairs. For example,

```
neg((3,4))
```

>> (-3,-4)

Similarly, after defining a midpoint operator

```
primarydef p mp q = 0.5[p,q] enddef;
```

we can write

```
z1 = (0,0);
z2 = (0,40);
draw z1 -- z2;
label.rt("mp", z1 mp z2);
```

mp
42

7 Macro Packages

Many METAPost macro packages have been written, by Hobby and others. In this section, we describe a few of them. To use a package, include this command at the beginning of your METAPost program:

```
input <package name>
```

7.1 Boxes

There is a standard macro package, called **boxes**, for drawing boxes. To use it, include this statement at the beginning of your METAPost program:

```
input boxes
```

The boxes produced by **boxes** have square corners. To get boxes with both square corners and rounded corners, use

```
input rboxes
```

7.1.1 Creating and Drawing Boxes

To create a box, write

```
boxit . <box name> ( <picture expression> )
```

The package **rboxes** provides the command **rboxit** as well. It works in the same way as **boxit**, but produces boxes with rounded corners.

To draw boxes, write

```
drawboxed (  $b_1, b_2, \dots, b_n$  )
```

where b_1, b_2, \dots, b_n is a list of box names. The result of drawing a box is shown in Figure 6, in which the shaded inner rectangle corresponds to the picture and the outer rectangle is the box that is drawn. The box can be sized and positioned by applying the suffixes in Figure 6 to the box name. For example, if the box name is **bx**, then **bx.c** is its centre, **bx.ne** is its top right corner, and so on.

The values of **c**, **dx**, and **dy** are left unspecified. If METAPost can infer their values from constraints on the positioning points, it does so. If not, it gives them the default values

```
c   = (0,0)
dx  = defaultdx
dy  = defaultdy
```

The values of **defaultdx** and **defaultdy** can be changed by assignment. For example:

```
defaultdx := 20;
defaultdy := 15;
```

The command **drawboxed** actually draws the boxes. Consequently, the best way to use boxes is to follow this sequence:

1. Use **boxit** to create the boxes
2. Write equations that specify positional relations between boxes
3. Use **drawboxed** to draw the boxes





Figure 6: Box variables


```
boxit.bl("left");
boxit.bm("middle");
boxit.br("right");
bm.sw - bl.ne = (20,0);
br.nw - bm.se = (20,0);
drawboxed(bl,bm,br);
```



30

This example is similar, but illustrates the use of the package `rboxes`.

```
boxit.sb("box with square corners");
rboxit.rb("box with rounded corners");
rb.c = sb.c + (100,0);
drawboxed(sb, rb);
cuta(sb,rb);
```




75

7.1.2 Positioning Boxes


METAPOST provides another, shorter way of positioning boxes relative to one another. The macro `boxjoin` takes as argument a list of equations, separated by semicolons, describing the relationship between boxes `a` and `b`. The equations can be used to constrain both the positions and the sizes of boxes.

```
boxjoin(a.sw = b.nw; a.se = b.ne);
boxit.bt("top");
bt.ne - bt.nw = (50,0);
boxit.bm("middle");
boxit.bb("bottom");
drawboxed(bt,bm,bb);
```




31

The equation `bt.ne - bt.nw = (50,0)` sets the width for the first box and `boxjoin` ensures that the other boxes have the same width. If it is omitted, the default size of the top box is used as the width:



The constraints imposed by `boxjoin` can be removed either by calling `boxjoin` with different equations or an empty argument.


```
boxjoin(a.e = b.w);
boxit.b1(btex $b_1$ etex); boxit.b2(btex $b_2$ etex);
boxit.b3(btex $b_3$ etex); boxit.b4(btex $b_4$ etex);
boxit.b5(btex $b_5$ etex);
boxjoin();
boxit.b6(btex $b_6$ etex); b6.n = b1.s - (whatever,10);
drawboxed(b1,b2,b3,b4,b5,b6);
```



36

The path surrounding the box named *b* is `bpath b`. This is useful when boxes are joined by arrows.

```
boxjoin(a.e = b.w - (20,30));
boxit.scan("scan");
boxit.parse("parse");
drawboxed(scan, parse);
drawarrow scan.c{dir 0} .. {dir 90}parse.c
    cutbefore bpath scan
    cutafter bpath parse;
```



33

In fact, this pattern occurs so often that it is a good idea to define a macro for it. To join two boxes `b1` and `b2` with an arrow, write `cuta(b1, b2)`, where:

```
vardef cuta(suffix a, b) =
    drawarrow a.c -- b.c cutbefore bpath.a cutafter bpath.b;
enddef;
```

7.1.3 Oval Boxes

The macro `circleit` is similar to `boxit` except that the picture argument is enclosed in an oval rather than a rectangle. The corner points are not defined, but the other points correspond to `boxit`, as shown in Figure 7. Use `drawboxed` to draw the picture produced by `circleit` (not “drawcircled”).

The path `bpath c` created by `circleit` is a circle unless at least one `c.dx`, `c.dy`, or `c.dx - c.dy` is known. If any of these values are known, the path is an oval that contains the picture with a safety margin defined by `circmargin`, a length with default value 2 bp.

Assigning to `defaultdx` and `defaultdy`, as explained above, does *not* work for `circleit`. However, their values can be assigned explicitly, as in this example:




Figure 7: Oval variables

```
circleit.wd("establishment");
wd.dy = 20;
drawboxed(wd);
```



35

Other commands associated with boxes include:

```
drawboxes(...) : draw the outlines of the listed boxes but not their contents
drawunboxed(...) : draw the contents of the listed boxes but not their outlines
pic b : return the contents of the box b without its outline
```

Thus `drawunboxed(b1,b2)` is an abbreviation for

```
draw pic b1;
draw pic b2;
```

7.2 Graphs

John Hobby has written a graph package for METAPost. A complete description of this package is beyond our scope, but we have space for a simple example. The file `hist.txt` contains the results of a simulated experiment in which a coin is tossed 50 times and the number of ‘heads’ are recorded. Each line of the file consists of two integers, as in the following extract:

```
....
23 96237
24 107870
25 112401
26 107747
27 96453
28 78692
....
```

Before drawing graphs, the graph macros must be read from `graph.mp`. The following METAPOST code is all that is needed to obtain Figure 8.

```
input graph

beginfig(1);
  draw begingraph(4in,3in);
  gdraw("hist.txt");
  endgraph;
endfig;
```




Figure 8: Simulated coin tossing

7.3 TEX

METAPOST does not process text between `btex` and `etex`. For example, if you write `btex n etex`, METAPOST will typeset the character “n”, not the value of the variable `n`. The package `TEX.mp` provides an operator `TEX` that overcomes this problem. The expression

```
TEX("$X_{" & decimal(n) & "}$")
```

returns a picture that can be used, for example, in a `label` command. For $n = 0, 1, 2, \dots$, the pictures will be X_0, X_1, X_2, \dots .

The package also provides two further commands:

<code>TEXPRE(s)</code>	defines text that will be passed to $\text{T}_{\text{E}}\text{X}$ before each <code>TEX</code> command
<code>TEXPOST(s)</code>	defines text that will be passed to $\text{T}_{\text{E}}\text{X}$ after each <code>TEX</code> command

For example, you can use \LaTeX in labels by including these two commands at the beginning of your METAPOST program (`char(10)` generates a line break):

```
TEXPRE("%&latex" & char(10) & "\documentclass{article}\begin{document}");
TEXPOST("\end{document}");
```

8 Debugging

It may seem odd to talk about “debugging” diagrams, but METAPOST is a programming language. The advantages of the programming language approach should be obvious from earlier parts of this manual; the disadvantages become clear when you actually try to use METAPOST.

As a simple illustration, suppose that you accidentally write “`btext`” instead of “`btex`”. If you have written a few hundred lines of code, METAPOST’s response may be quite worrying:

```
D:\Pubs\MetaPost>mp figs
(figs.mpCreating figs.mpx...
makempx: mpto failed on D:\Pubs\MetaPost\figs.mp.
mp: The operation failed for some reason.
```

As with other kinds of programming, the best approach to METAPOST is to proceed slowly and carefully with frequent tests. When METAPOST does fail, there is a good chance that the error is in the last few lines that you have written.

Here are a few additional tips for getting errant METAPOST programs to work:

- Use `show` and its friends (§5.2.8:35) to check that variables have the values that you expect.
- Use commands like `dotlabel("z5", z5)` to check that points are where you think they are.
- When METAPOST fails, it often generates hundreds of lines of diagnostics. Sometimes, it is easy to see what went wrong. If not, it is often quicker to identify the line of code that METAPOST is complaining about, and to examine that line very carefully, than it is to try and understand the diagnostic.
- Errors such as “inconsistent equation” may occur if you use the same variable names in different figures. You can avoid this problem by using `save` (§5.2.7:35) for the variables of each figure.

9 Examples

In the last section of this manual, we give some complete examples that illustrate the practical use of METAPOST.

9.1 Euler Integration

The goal of this example is to obtain a figure like Figure 9 illustrating the error, e , of first-order Euler integration. If f' is the derivative of f :

$$f(x + \Delta x) \approx f(x) + f'(x) \Delta x$$




Figure 9: Euler Integration

The first step is to draw a suitable path. A Bézier curve is the simplest to generate with METAPOST:

```
path c;
c = (0,0)dir 0 .. (150,100);
draw c;
```

Next, we choose a point on the path, P_1 :

```
t := 0.4;
pair P[];
P1 = point t of c;
```

The value of Δx is arbitrary, but we should choose a value large enough to make the diagram readable. We can then compute Δy using the direction (i.e., slope) of the path at t .

```
dx := 60;
(dx, dy) = whatever * direction t of c;
```

With Δx and Δy known, we can locate the other points that we need. P_4 is not labelled in Figure 9; it is the third vertex of the triangle with hypotenuse P_1P_3 .

```
P4 = P1 + (dx, 0);
P3 = P4 + (0, dy);
P2 = c intersectionpoint (P4 -- P4+(0,100));
```

All that remains is to draw the lines and label the points and lengths:

```
draw P1--P4--P2;
draw P1--P3;
dotlabel.ulft(btex $P_1$ etex, P1);
dotlabel.ulft(btex $P_2$ etex, P2);
dotlabel.rt(btex $P_3$ etex, P3);
```



```
label.bot(btex $\Delta x$ etex, 0.5[P1,P4]);
label.rt(btex $\Delta y$ etex, 0.5[P4,P3]);
label.rt(btex $e$ etex, 0.5[P3,P2]);
```

9.2 The Lorentz Transformation

The Lorentz transformation for inertial frames with relative velocity v is

$$\begin{aligned}x' &= \gamma x + \gamma \beta ct \\ ct' &= \gamma \beta x + \gamma ct\end{aligned}$$

where c is the velocity of light, $\beta = v/c$, and $\gamma = 1/\sqrt{1 - v^2/c^2}$. Assuming $c = 1$, $v = 0.5$, writing b for β , g for γ , and including a translation of 200 units, we can express the transformation in METAPOST as:

```
b = 0.5;
g = sqrt(1 - b * b);
transform t;
xpart t = 200; ypart t = 0;
xxpart t = g; yxpart t = g * b;
xypart t = g * b; yypart t = g;
```

We illustrate the scenario shown on the left of Figure 10): a light flashes at a , is reflected by mirrors at b and b' , and returns to the original point, displaced in time, at a' . We need eleven points for the diagram; Figure 11 shows the numbering scheme. The scale is determined by two constants, s and d :

```
s = 10;
d = 80;
```

It is easiest to define the x coordinates and the y coordinates separately:

```
x1 = x2 = x3 = 0;
x4 = x5 = x6 = x7 = x1 + d;
x8 = x9 = x10 = x4 + d;
x11 = x8 + s;
y1 = y4 = y8 = y11 = 0;
y5 = s;
y2 = y9 = y5 + d;
y6 = y2 + d;
y3 = y7 = y10 = y6 + s;
```

We draw arrows for the axes:

```
drawarrow z1 -- z11;
drawarrow z1 -- z3;
```

solid lines for the time lines:

```
draw z1 -- z3;
draw z4 -- z7;
draw z8 -- z10;
```




Figure 10: The Lorentz Transformation

3	7	10
	6	
2		9
1	5	8 11
	4	

Figure 11: Point numbering used in Figure 10

and dashed lines for the light rays:

```
draw z5 -- z2 dashed evenly;
draw z2 -- z6 dashed evenly;
draw z5 -- z9 dashed evenly;
draw z9 -- z6 dashed evenly;
```

These commands have drawn into `currentpicture`. We store them in the picture variable `pic` and clear the current picture using the internal constant `nullpicture`:

```
picture pic;
pic := currentpicture;
currentpicture := nullpicture;
```

We next draw `p` twice, once as above, and once transformed:

```
draw pic;
draw pic transformed t;
```

We then label the original diagram:

```
label.llft(btex $O$ etex, z1);
label.rt(btex $X$ etex, z11);
label.top(btex $T$ etex, z3);
label.rt(btex $a$ etex, z5);
label.lft(btex $b$ etex, z2);
label.rt(btex $b'$ etex, z9);
label.rt(btex $c$ etex, z6);
```

and the transformed diagram:

```
label.llft(btex $O$ etex, z1 transformed t);
label.rt(btex $X$ etex, z11 transformed t);
label.top(btex $T$ etex, z3 transformed t);
label.rt(btex $a$ etex, z5 transformed t);
label.lft(btex $b$ etex, z2 transformed t);
label.rt(btex $b'$ etex, z9 transformed t);
label.rt(btex $c$ etex, z6 transformed t);
```

The result is Figure 10. Note that the events b and b' are simultaneous in the original diagram but not in the transformed diagram, and that the light rays (dashed lines) have the same slope (± 1) in both diagrams.

This example would be slightly simpler if we included the labels in the first diagram. Then we would not have to write them out twice. Since METAPOST transforms *everything*, however, the labels would be distorted.

9.3 Dissipation

Figure 12 shows how the area between two trajectories of a dissipative system diminishes when measured at constant time intervals. The code for drawing it uses `point...of` to obtain points on the upper and lower curves, and `buildcycle` to construct the areas for shading.




Figure 12: Area reduction in a dissipative system

9 Examples

Pick four points as end points for the curves:

```
z1 = (0,100); z2 = (160,0);  
z3 = (0,60); z4 = (150,0);
```

Draw two curves, converging towards each other:

```
path p[];  
p1 = z1{dir 0} .. {dir 270}z2;  
p2 = z3{dir 0} .. {dir 270}z4;
```

Construct pairs of lines joining corresponding points on the curves. The lines are not drawn, but are used as arguments of `buildcycle` to create closed areas, which are shaded with `filldraw`.

```
for i = 0 step 2 until 11:  
  p3 := point i/11 of p1 -- point i/11 of p2;  
  p4 := point (i+1)/11 of p1 -- point (i+1)/11 of p2;  
  filldraw buildcycle(p3, p1, p4, p2) withcolor 0.7 white;  
endfor;
```

Finally draw the curves `p1` and `p2`, and a couple of lines, to bound the figure.

```
draw p1; draw p2;  
draw z1 -- z3; draw z2 -- z4;
```

9.4 Desargues' Theorem

Figure 13 is obtained by the following construction.

- Choose a centre of projection, O .
- Draw a triangle, $A_1B_1C_1$.
- Draw the lines OA_1 , OA_2 , and OA_3 .
- Draw another triangle, $A_2B_2C_2$, with the same centre of projection. That is A_2 lies on OA_1 , etc.

In METAPOST, this step is done by mediation. E.g., `a2 = 0.35[a1,o]`.

- Find the points of intersection of corresponding sides of the triangles. For example, B_1C_1 and B_2C_2 intersect at P_1 .

In METAPOST, this step is also done by mediation. E.g.:

```
p1 = whatever[b1,c1] = whatever[b2,c2];
```

- Desargues' Theorem states that the three points of intersection, P_1 , P_2 , and P_3 , are collinear.

Figure 14 shows the METAPOST code used to draw Figure 13. The label commands have been put at the end to ensure that the black dots are drawn after the coloured lines that go through them.




Figure 13: Desargue's Theorem

9.5 Three Dimensions

Although METAPost is basically two-dimensional, it can be tricked into drawing perspective diagrams, giving the illusion of three dimensions. Many packages have been written for 3D METAPost pictures (§10:53). One technique is to use the three components of the type `color` as XYZ coordinates. We give a simple example to illustrate the idea.

Assume that the viewpoint in 3D space is at $(0, 0, -d)$ and that we project onto the plane $z = 0$. This projection maps the 3D point (x, y, z) to the 2D point (tx, ty) where $t = d/(z + d)$. The macro `perspective` implements this transformation by converting a colour $c = (r, g, b)$ to a pair (tr, tg) :

```
vardef perspective(expr c) =
  t := dist/(bluepart c + dist);
  (t * redpart c, t * greenpart c)
enddef;
```

We use the colour array `tp[]` to represent the vertexes of a cube with side 2 centered at the origin:

```
color tp[];
tp0 = (-1,-1,-1); tp4 = tp0 + (0,0,2);
```

```

pair a[],b[],c[],p[], o;
o = (300,0);

a1 = (0,0); b1 = (60,135); c1 = (95,20);

draw a1 -- b1 -- c1 -- cycle withcolor red;

draw a1 -- o withcolor blue;
draw b1 -- o withcolor blue;
draw c1 -- o withcolor blue;

a2 = 0.35[a1,o]; b2 = 0.2[b1,o]; c2 = 0.5[c1,o];

draw a2 -- b2 -- c2 -- cycle withcolor green;

p1 = whatever[b1,c1] = whatever[b2,c2];
p2 = whatever[c1,a1] = whatever[c2,a2];
p3 = whatever[a1,b1] = whatever[a2,b2];

draw b1 -- p1 -- b2;
draw a1 -- p2 -- a2;
draw b1 -- p3 -- b2;
draw p1 -- p2 -- p3 dashed evenly;

dotlabel.rt(btex $O$ etex, o);
dotlabel.ulft(btex $A_1$ etex, a1);
dotlabel.lft(btex $B_1$ etex, b1);
dotlabel.bot(btex $C_1$ etex, c1);

dotlabel.bot(btex $A_2$ etex, a2);
dotlabel.llft(btex $B_2$ etex, b2);
dotlabel.urt(btex $C_2$ etex, c2);

dotlabel.lft(btex $P_1$ etex, p1);
dotlabel.bot(btex $P_2$ etex, p2);
dotlabel.top(btex $P_3$ etex, p3);

```

Figure 14: Programming Desargues' Theorem

```

tp1 = ( 1,-1,-1); tp5 = tp1 + (0,0,2);
tp2 = ( 1, 1,-1); tp6 = tp2 + (0,0,2);
tp3 = (-1, 1,-1); tp7 = tp3 + (0,0,2);

```

Since METAPOST supports linear operations on colours, we can scale and translate the cube before using perspective with `dist = 200` to transform the vertexes to the pair array `p[]`:

```
dist := 200;
```

```

pair p[];
for i = 0 upto 7:
    p[i] := perspective(40*tp[i]+(80,-70,30));
endfor;

```

Drawing the edges of the cube gives Figure 15:

```

draw p0--p1--p2--p3--cycle;
draw p4--p5--p6--p7--cycle;
draw p0--p4;
draw p1--p5;
draw p2--p6;
draw p3--p7;

```




Figure 15: Perspective view of a cube

9.6 Reading a File

This example illustrates the use of `readfrom` and `scantokens`. Suppose that the file "data.txt" contains these two lines:

```

1, 2, x, "Go!"
"pi = ", 3.14159263

```

The code on the left produces the figure on the right:

```

boxjoin(a.sw = b.nw; a.se = b.ne);
picture b[];
i = 0;
x = 99;
forever:
  string line;
  line := readfrom "data.txt";
  exitif line = EOF;
  string token, desc;
  for token = scantokens line:
    if numeric token:
      desc := "Numeric: " &
        decimal token;
    elseif string token:
      desc := "String: " & token;
    fi;
    boxit.b[i](desc);
    b[i].e = b[i].w + (100,0);
    drawboxed(b[i]);
    i := i + 1;
  endfor;
endfor;

```

Numeric: 1
Numeric: 2
Numeric: 99
String: Go!
String: pi =
Numeric: 3.14159

72

The outer `forever` loop reads each line of the file, checking for end of file. The inner `for` loop uses `scantokens` to split the line into tokens. The type of each token is checked, using type names as predicates, and put into a descriptor, `desc`. Finally, each descriptor is drawn in a box. Note that the third token in the file is `x` and it is displayed as `99`, the value of `x` in the program. Tokens in a file are considered to be strings only if they are quoted, like `"Go!"`.

10 Links

The T_EX User's Group — probably all that you will need: <http://www.tug.org/metapost.html>

A FAQ from the UK: <http://www.tex.ac.uk/cgi-bin/texfaq2html?label=MP>

Urs Oswald's *Very Brief Tutorial*: <http://www.ursoswald.ch/metapost/tutorial.html>

METAPOST home page: <http://ect.bell-labs.com/who/hobby/MetaPost.html>

John Hobby's home page: <http://plan9.bell-labs.com/who/hobby/>

Marc van Dongen's contributions to METAPOST: <http://www.cs.ucc.ie/~dongen/mpost/mpost.html>

MetaFun by Hans Hagen: <http://wiki.contextgarden.net/MetaFun>

A WYSIWYG interface for METAPOST: <http://w3.mecanica.upm.es/metapost/metagraf.php>

A METAPOST generator for 3D: <http://www.gnu.org/software/3dldf/LDF.html>

Index

- [&](#), [11](#), [21](#), [27](#)
- [**](#), [8](#)
- [++](#), [8](#), [13](#)
- [+-+](#), [8](#), [13](#)
- [--](#), [21](#)
- [---](#), [23](#)
- [. .](#), [23](#)
- [. . .](#), [23](#)
- [:=](#), [8](#)
- [=](#), [8](#)
- [\[\]](#), [14](#)
- [3D](#), [50](#)

- [abs](#), [13](#), [14](#)
- [affine transform](#), [17](#)
- [ahangle](#), [30](#)
- [ahlength](#), [30](#)
- [and](#), [9](#)
- [angle](#), [14](#)
- [angle](#), [14](#), [18](#)
- [anonymous variable](#), [9](#)
- [arc](#), [22](#)
- [arclength](#), [26](#)
- [arctime](#), [8](#), [26](#)
- [area](#)
 - [shaded](#), [27](#)
- [argument](#), [36](#)
- [arithmetic](#), [12](#)
- [array](#), [6](#)
- [arrow](#), [30](#)
- [assignment](#), [8](#)

- [background](#), [12](#)
- [bbox](#), [26](#), [27](#)
- [beginfig](#), [14](#), [32](#)
- [beginfig](#), [2](#)
- [begingroup](#), [6](#)
- [beveled](#), [15](#)
- [big points](#), [4](#)
- [binary operator](#), [7](#)
- [black](#), [11](#)
- [blackpart](#), [12](#)
- [blue](#), [11](#)
- [bluepart](#), [12](#)
- [boolean](#), [9](#), [10](#)
- [bot](#), [31](#)

- [bounded](#), [20](#), [33](#)
- [bounding box](#), [19](#)
- [boxes](#), [39](#)
- [boxit](#), [39](#)
- [boxjoin](#), [40](#)
- [bp \(big point\)](#), [4](#)
- [bpath](#), [41](#)
- [btex](#), [28](#), [43](#)
- [buildcycle](#), [27](#), [48](#)
- [butt](#), [15](#)

- [c](#), [39](#)
- [ceiling](#), [13](#)
- [center](#), [19](#)
- [char](#), [10](#)
- [circle](#), [22](#)
- [circleit](#), [41](#)
- [circmargin](#), [41](#)
- [clip](#), [29](#)
- [clipped](#), [20](#), [33](#)
- [cm](#), [4](#)
- [cmykcolor](#), [10](#), [12](#)
- [color](#), [10](#), [11](#), [50](#)
- [comments](#), [4](#)
- [comparison](#), [9](#)
- [concatenation](#)
 - [of paths](#), [21](#)
 - [of strings](#), [11](#)
- [constants](#)
 - [colours](#), [11](#)
 - [numerics](#), [12](#)
 - [pairs](#), [14](#)
 - [strings](#), [10](#)
- [control points](#), [24](#)
- [conversion](#)
 - [abs \(pair to numeric\)](#), [14](#)
 - [angle \(pair to numeric\)](#), [14](#)
 - [ceiling \(numeric to integer\)](#), [13](#)
 - [char \(ASCII to string\)](#), [10](#)
 - [decimal \(numeric to string\)](#), [10](#)
 - [dir \(numeric to pair\)](#), [14](#)
 - [floor \(numeric to integer\)](#), [13](#)
 - [round \(numeric to integer\)](#), [13](#)
 - [str \(suffix to string\)](#), [11](#)
- [cosd](#), [13](#)

- curl, 23
- currentpicture, 5, 19, 47
- curve, 23
- cuta(b1, b2), 41
- cutafter, 27
- cutbefore, 27
- cyanpart, 12
- cycle, 10, 21, 26

- dashed, 31
- dashed line, 22
- dashpart, 20, 33
- day, 12
- decimal, 10, 13
- declaration, 6
- def, 36
- defaultdx, 39
- defaultdy, 39
- defaultfont, 11, 28
- defaultpen, 15
- defaultscale, 11, 28
- difference of squares, 8, 13
- dir, 14
- direction, 14
 - of curve, 23
 - represented by pair, 14
- direction, 8, 18, 26
- directionpoint, 8, 26
- directiontime, 8, 26
- ditto, 10
- div, 13
- division, 13
- dotlabel, 31
- dotlabels, 31
- dotprod, 14
- dotted line, 22
- down, 14
- downto, 21, 33
- draw, 12, 29
- drawarrow, 30
- drawboxed, 39
- drawboxes, 42
- drawdblarrow, 30
- drawoptions, 12, 16, 31
- drawunboxed, 42
- dx, 39
- dy, 39
- else, 33
- elseif, 33
- endfig, 2
- endgroup, 6
- EOF, 35, 36
- epsilon, 12
- equality, 8, 9
- equation, 8
- erase, 29
- etex, 43
- evenly, 22
- exitif, 33
- exitunless, 33
- explicit declaration, 6
- expr
 - macro parameter, 37
- expression, 7
 - path, 26

- false, 9
- fi, 33
- file, 2
 - name, 32
 - read, 35, 52
 - write, 36
- filenametemplate, 2, 32
- fill, 12, 30
- filldraw, 12, 27, 30
- filled, 20, 33
- floor, 13
- font, 28
- for, 21, 32, 53
- forever, 33, 53
- fullcircle, 22, 26
- functions
 - colours, 11
 - numerics, 13
 - pairs, 14
 - pictures, 19
 - strings, 10

- green, 11
- greenpart, 12
- group, 6

- halfcircle, 22
- hypotenuse, 8, 14

- identity, 17
- if, 33

- image, 20
- in, 4
- infinity, 12
- inflection points, 23
- infont, 11
- inner product, 14
- input, 38
- input file, 2
- interim, 7
- internal variable, 5
- intersection, 25
- intersectionpoint, 26
- intersectiontimes, 26
- inverse, 17

- known, 10

- label, 31, 43
- L^AT_EX, 28
- left, 14
- length, 12
- length, 10, 19, 26
- lft, 31
- line
 - corner, 15
 - end, 15
 - straight, 21
- linecap, 15
- linejoin, 15
- list, 20
- llcorner, 19
- llft, 31
- local
 - variable, 6
- loop, 32
- lrcorner, 19
- lrt, 31

- macro, 36
- magentapart, 12
- makepath, 15
- makepen, 15
- mediation, 13, 14
- message, 34
- mexp, 13
- mitered, 15
- miterlimit, 15
- mlog, 13
- mm, 4

- mod, 13
- month, 12
- mproof, 3
- mpversion, 2, 10
- multidimensional array, 6

- name, 5
 - local, 6
 - of output file, 32
 - of type, 6
 - of variable, 5
- negative coordinates, 5
- newinternal, 5
- nib, 15
- normal, 18
 - random number, 13
- normaldeviate, 13
- normalize, 14
- not, 9
- nullpen, 15
- nullpicture, 19, 47
- numeric, 10, 12, 13

- odd, 10, 13
- of, 8
- operator, 7
 - assignment, 8
 - comparison, 9
 - difference of squares, 8
 - equality, 8
 - path, 21, 23
 - sum of squares, 8
- or, 9
- origin, 14
- output
 - file name, 32
 - to console, 34, 35
 - to file, 2

- painting, 30
- pair, 10, 13
- parameter, 36
 - of macro, 37
- path
 - circle, 22
 - dots and dashes, 22
 - expression, 26
 - of box, 41
 - parametric, 24

- straight, 21
- path, 10, 21, 26
- pathpart, 20, 33
- pc, 4
- pen, 14
- pen, 10
- pencircle, 15
- penoffset, 8
- penpart, 20, 33
- pensquare, 15
- pic, 42
- pickup, 14
- picture, 19
- picture, 10
- point
 - control, 24
 - Postscript, 4
 - printer's, 4
- point, 8, 18, 26, 48
- postcontrol, 8
- precedence, 7
- precontrol, 8
- predicates, 10
- primary, 7
- pt (printer's point), 4
- quartercircle, 22
- random number, 13
- rboxes, 39, 40
- rboxit, 39
- read file, 35, 52
- readfrom, 35, 52
- red, 11
- redpart, 12
- reflectedabout, 17
- remainder, 13
- reverse, 26, 27
- rgbcolor, 10, 12
- right, 14
- rotated, 17
- rotatedaround, 17
- round, 13, 14
- rounded, 15
- rt, 31
- save, 6, 35
- scaled, 17, 22
- scaling, 4
- scantokens, 10, 21, 33, 52
- scope, 6
- secondary, 7
- semicircle, 22
- setbounds, 20
- shading, 27, 30
- shadow, 36
- shifted, 17, 22
- show, 35
- showdependencies, 9, 35
- showstopping, 35
- showtoken, 35
- showvariable, 35
- sind, 13
- slanted, 17
- sqr, 13
- squared, 15
- step, 21
- str, 11
- string, 10
- string, 10
- stroked, 20, 33
- subpath, 8, 22, 27
- substring, 8, 10
- suffix, 5
- suffix
 - macro parameter, 37
- sum of squares, 8, 13
- tag, 5
- tangent, 18
- tension, 24
- tertiary, 7
- TEX, 43
- TEX, 28
- text
 - macro parameter, 37
- textual, 20, 33
- thelabel, 31
- three dimensions, 50
- time, 12
- top, 31
- transform, 17
- transform, 10, 17
- transformed, 17
- true, 9
- ulcorner, 19
- ulft, 31

Index

UMM, 1
unary operator, 7
undefine, 6
undraw, 29
unfill, 30
uniform random number, 13
uniformdeviate, 13
units, 4
unitsquare, 26, 27
unitvector, 14
unknown, 10
until, 21
up, 14
upto, 21, 33
urcorner, 19
urt, 31

value
 colour, 11
 decimal, 10
 global, 6
 group, 6
 known, 10
 largest, 12
 smallest, 12
 unknown, 6, 10
vardef, 38
variable
 anonymous, 9
 declaration, 6
 internal, 5, 7
 local, 6
 name, 5
verbatimtex, 28
version, 2

whatever, 9
white, 11
withcolor, 12, 30, 31
withdots, 22
within, 33
withpen, 16, 31
write
 expression, 35
 string, 34
 to file, 36
write, 36

x, 5, 14, 38
xpart, 14, 18
xscaled, 15, 17
xxpart, 18
xypart, 18

y, 5, 14, 38
year, 12
yellowpart, 12
ypart, 14, 18
yscaled, 15, 17
yxpart, 18
yypart, 18

z, 5, 14, 38
zscaled, 17